Singularity Loci of 3-RRR Planar Parallel Manipulators General Case: Distal and Proximal Links of Different Lengths

Ilian A. Bonev and Clement M. Gosselin October 24, 2000

> restart: with(linalg):

Part I: Deriving the Expre ssion for the Jacobian Matrix

In this case we cannot proceed symbolically but instead assign random integer numerical values to the parameters describing the parallel manipulator.

> RandNum := evalf(rand(-13..12)):
  for i from 1 to 3 do
    a||i := RandNum(): b||i := RandNum():
    c||i := RandNum(): d||i := RandNum():
  od:
  l1 := 1: l2 := 1+abs(RandNum()):
  a1 := 0: b1 := 0: c1 := 0: d1 := 0:
  d2 := b2:

1. Unit Vectors Along Distal Links

In all the derivations, we consider only one branch.

> for i from 1 to 3 do
    p||i := (rho||i + l1^2 - l2^2)/(2*l1):
    Gamma||i := rho||i - p||i^2:
    sinT||i := (p||i*(Y+b||i) + (X+a||i)*Delta||i)/rho||i:
    cosT||i := (p||i*(X+a||i) - (Y+b||i)*Delta||i)/rho||i:
    rho||i := (X+a||i)^2 + (Y+b||i)^2:
    n||i := [X + a||i - l1*cosT||i, Y + b||i - l1*sinT||i]:
  od:

2 . Vectors E*si

> for i from 1 to 3 do
    es||i := [-d||i, c||i];
  od:

3 . Jacobian Matrix

> J := matrix(3,3,[n1[1],n1[2],n1[1]*es1[1] + n1[2]*es1[2],
                   n2[1],n2[2],n2[1]*es2[1] + n2[2]*es2[2],
                   n3[1],n3[2],n3[1]*es3[1] + n3[2]*es3[2]]):

Part II: Obtaining the Singularity Locus

> Eq := simplify(numer(det(J))):

1 . Getting Rid of The First Square Root

After rearranging Eq and raising to power two, we cancel the common factors rho1, rho2, and rho3. Note that these comprise in fact the numerator of the determinant of J.

> Eq1 := expand(coeff(Eq,Delta1,1)^2*Gamma1 - coeff(Eq,Delta1,0)^2):
  Eq1 := factor(Eq1/rho1):
  Eq1 := expand(subs(Delta3^2=Gamma3, subs(Delta2^2=Gamma2, Eq1) )):

2 . Getting Rid of The Second Square Root

> Eq2 := expand(coeff(Eq1,Delta2,1)^2*Gamma2) - expand(coeff(Eq1,Delta2,0)^2):
  Eq2 := factor(Eq2/rho2^2):
  Eq2 := simplify(subs(Delta3^2=Gamma3,Eq2)):

3 . Getting Rid of The Third Square Root

> Eq3 := expand(coeff(Eq2,Delta3,1)^2*Gamma3) - expand(coeff(Eq2,Delta3,0)^2):
  Eq3 := factor(Eq3/rho3^4):
> degree(Eq3,X), degree(Eq3,Y), degree(Eq3,[X,Y]);

, , 42 42 42